

Inhaltsverzeichnis

1	Motivation.....	2
2	Der "übliche" Weg.....	2
2.1	Video soll interlaced bleiben.....	2
2.1.1	Ohne temporale Filter (z.B. nur Resizen u. Farbanpassungen).....	2
2.1.2	Mit temporalen Filtern (z.B. Rauschfilter)	2
2.2	Video soll deinterlaced (progressive) werden	2
3	Filtern von interlaced Video mittels Bobber	3
3.1	Verwendete Funktionen.....	3
3.2	Video soll interlaced bleiben.....	4
3.2.1	Saubere (gering verrauschte) Quelle	4
3.2.2	Stark verrauschte Quelle (z.B. analog Capture).....	4
3.3	Video soll deinterlaced (progressive) werden	5
3.3.1	Saubere (gering verrauschte) Quelle	5
3.3.2	Stark verrauschte Quelle (z.B. analog Capture).....	5
4	Danksagung.....	5

1 Motivation

Der "übliche" Weg interlaced Video in AviSynth zu filtern ist in Kapitel 2 *Der "übliche" Weg* dargestellt. Unter Verwendung eines Bobbers können jedoch wesentlich bessere Ergebnisse erzielt werden. In Kapitel 3 *Filtern von interlaced Video mittels Bobber* werden je nach Art des Quellmaterials (sauber od. verrauscht), des gewünschten Endergebnisses (interlaced od. progressive) und der verwendeten Filter (nur spatial od. auch temporal) die jeweiligen Wege gezeigt.

2 Der "übliche" Weg

2.1 Video soll interlaced bleiben

2.1.1 Ohne temporale Filter (z.B. nur Resizen u. Farbanpassungen)

```
AviSource("InterlacedVideo.avi",true,"YUY2")

SeparateFields()

# spatiale Filter
BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(50)

Weave()
```

2.1.2 Mit temporalen Filtern (z.B. Rauschfilter)

```
AviSource("InterlacedVideo.avi",true,"YUY2")

SeparateFields()

# temporale Filter
tfield = SelectOdd().UnDot().PeachSmoother()
bfield = SelectEven().UnDot().PeachSmoother()
Interleave(tfield,bfield)

# spatiale Filter
BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(50)

Weave()
```

2.2 Video soll deinterlaced (progressive) werden

```
AviSource("InterlacedVideo.avi",true,"YUY2")

TomsMoComp(-1,5,0)

UnDot().PeachSmoother()
BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(50)
```

3 Filtern von interlaced Video mittels Bobber

3.1 Verwendete Funktionen

Im weiteren Verlauf werden folgende Funktionen verwendet:

```
#####
# TMCBob(clip a, bool "tff")
#
# Erzeugt aus interlaced Video ein progressives doppelter Framerate
# (z.B. aus 25 fps interlaced ein 50 fps progressive)
#
# tff gibt die Field order des clip a an:
# true --> Top Field First (Upper Field First)
# false --> Botttom Field First (Lower Field First)
# Falls nicht angegeben, wird true (Top Field First) angenommen
#####

#####
# ReinterlaceBob(clip a, bool tff)
#
# Erzeugt aus einem progressiven ein interlaced Video halber Framerate
# (z.B. aus 50 fps progressive ein 25 fps interlaced)
#
# tff gibt die Field order des zu erzeugenden interlaced Videos an:
# true --> Top Field First (Upper Field First)
# false --> Botttom Field First (Lower Field First)
# Falls nicht angegeben, wird true (Top Field First) angenommen
#####

#####
# HalfFpsBob(clip a)
#
# Erzeugt aus einem via TMCBob auf doppelte Framerate gebobbtten Video
# ein progressives halber Framerate
# (z.B. aus 50 fps progressive ein 25 fps progressive)
#####

function TMCBob(clip a, bool "tff") {
    tff = Default(tff,true)
    a = tff ? a.AssumeTFF() : a.AssumeBFF()
    f = a.TomsMoComp(-1,5,0)
    e = a.DoubleWeave().SelectOdd().TomsMoComp(-1,5,0)
    Interleave(f,e).AssumeFrameBased()
}

function ReinterlaceBob(clip a, bool "tff") {
    tff = Default(tff,true)
    a = tff ? a.AssumeTFF() : a.AssumeBFF()
    a = a.ConvertToYUY2()
    SeparateFields(a).SelectEvery(4,0,3).Weave()
}

function HalfFpsBob(clip a) {
    SelectOdd()
}
```

¹⁾ Bug in AviSynth: AviSource() setzt immer auf BFF, auch wenn das AVI tats. TFF ist. Daher wird in TMCBob() ggf. wieder auf TFF korrigiert.

²⁾ Bug in AviSynth: Bob() und AssumeFrameBased() setzen intern auf BFF, auch wenn die Quelle tats. TFF ist. Daher wird in ReinterlaceBob() ggf. wieder auf TFF korrigiert.

3.2 Video soll interlaced bleiben

3.2.1 Saubere (gering verrauschte) Quelle

```
AviSource("InterlacedVideo.avi",true,"YUY2")

TMCBob() # bzw. TMCBob(tff=false)

UnDot().TemporalCleaner(3,6)
BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(100)

ReinterlaceBob() # bzw. ReinterlaceBob(tff=false)
```

3.2.2 Stark verrauschte Quelle (z.B. analog Capture)

```
AviSource("InterlacedVideo.avi",true,"YUY2")

Bob()
UnDot().PeachSmoother()
ReinterlaceBob() # bzw. ReinterlaceBob(tff=false)

TMCBob() # bzw. TMCBob(tff=false)

BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(50)

ReinterlaceBob() # bzw. ReinterlaceBob(tff=false)
```

3.3 Video soll deinterlaced (progressive) werden

3.3.1 Saubere (gering verrauschte) Quelle

```
AviSource("InterlacedVideo.avi",true,"YUY2")

TMCBob() # bzw. TMCBob(tff=false)

UnDot().TemporalCleaner(3,6)
BicubicResize(480,352,0,0.5)
ColorYUV(levels="PC->TV")
AddBorders(16,16,16,16)
FadeOut(100)

HalfFpsBob()
```

3.3.2 Stark verrauschte Quelle (z.B. analog Capture)

Wenn nach dem Entrauschen keine temporalen Filter mehr eingesetzt werden:

```
AviSource("InterlacedVideo.avi",true,"YUY2")

Bob()
UnDot().PeachSmoother()
ReinterlaceBob() # bzw. ReinterlaceBob(tff=false)

TomsMoComp(-1,5,0)

BicubicResize(480,352,0,0.5) # Hier keine temporalen Filter!
ColorYUV(levels="PC->TV") #
AddBorders(16,16,16,16) #
FadeOut(50) #
```

bzw. falls nach dem Entrauschen auch temporale Filter eingesetzt werden sollen:

```
AviSource("InterlacedVideo.avi",true,"YUY2")

Bob()
UnDot().PeachSmoother()
ReinterlaceBob() # bzw. ReinterlaceBob(tff=false)

TMCBob() # bzw. TMCBob(tff=false)

IrgendeinTemporalerFilter() # Hier sind auch temporale Filter zulässig
BicubicResize(480,352,0,0.5) #
ColorYUV(levels="PC->TV") #
AddBorders(16,16,16,16) #
FadeOut(100) #

HalfFpsBob()
```

4 Danksagung

Obiges Vorgehen entstammt den Vorschlägen von scharfis_brain (<http://forum.gleitz.info>) bei welchem ich mich dafür herzlich bedanken möchte!